Appendix

A1. Data Preparation

```r
# R packages
library(ggplot2)
library(tidyverse)
library(Hmisc)
library(PMXForest)
library(patchwork)
library(kableExtra)
library(ggpubr)
library(scales)
```

```r
# data preparation
data <- read.table("dataset.txt", sep = " ",       # space-separated
                   header = TRUE, na.strings = ".") # treat "." as NA

LLOQ <- 1 # mg/L

data <- data %>% mutate(DOSE = as.numeric(DOSE),
                        EVID = case_when(is.na(CONC) & DOSE!=0 ~ 1,
                                         CONC>=0 ~ 0,
                                         is.na(CONC) & is.na(DOSE) ~ 2),
                        BLQ = case_when(CONC<1 ~ 1,
                                        CONC>=1 ~ 0,
                                        is.na(CONC) ~ 0),
                        CONC_M7 = ifelse(CONC < 1, 0, CONC)) %>%
  group_by(ID) %>%
  fill(c(ALB, BW, CREAT),.direction = "downup") %>%
  arrange(TIME, .by_group = TRUE) %>%
  mutate(ADA = case_when(
    ID %in% 71:75 ~ as.integer(cumany(ADA == 1)),
    TRUE          ~ ADA),
         COL = if_else(cummax(COL == 1L) == 1L, 1L, COL)) %>%
  ungroup()

data$DOSE = data$DOSE * data$BW # dose (mg) = dose (mg/kg) * body weight (kg)

write.csv(data, file = "dataset_clean.csv", quote = FALSE,
          na = ".", row.names = FALSE)
```

A2. Final Model

A2.1. Two-Compartment Target-Mediated Drug Disposition Model with Michaelis-Menten Approximation

```
$PROBLEM PMX001 Two Compartment TMDD Model
;; 2 assumptions:
;; 1. Free ligand concentration is much larger than total target conc.
;; 2. Vmax = cst => Rtot = cst => target degradation rate (kout)
;;    = ligand-target complex internalization rate (kint)

$INPUT DUMMY=DROP ID TIME WEEK DOSE=AMT RATE CONC=DROP ADA BW ALB SEX AGE
       CREAT COL EVID BLQ CONC_M7=DV

$DATA dataset_clean.csv IGNORE=@

$ABBR DERIV2=NO

$SUBROUTINES
ADVAN13
TOL=5

$MODEL
NCOMP=3                       ; number of compartments
COMP=(DOSE, DEFDOSE, DEFOBS) ; first (central) compartment
COMP=(PERIPH)                 ; second (peripheral) compartment
COMP=(AUC)                    ; continuous AUC integration

$PK
IF (EVID.EQ.1) LASTDOSE = TIME
TAD = TIME - LASTDOSE

  TVCL = THETA(1)
TVVMAX = THETA(2)
  TVKM = THETA(3)
  TVV1 = THETA(4)
  TVV2 = THETA(5)
   TVQ = THETA(6)

CLALB = ((ALB/31.85)**THETA(7))

  CL = TVCL   * EXP(ETA(1)) * CLALB
VMAX = TVVMAX
  KM = TVKM
```

```
  V1 = TVV1
  V2 = TVV2
   Q = TVQ

 K10 = CL/V1
 K12 = Q/V1
 K21 = Q/V2

  S1 = V1/1 ; scale prediction based on DOSE (mg) and DV (mg/L)
  S2 = V2/1

$THETA ; values are determined in 3 iterations
(0.01,     0.944,       2.3) ; CL (L/d)
(0.5,      1.38,          3) ; VMAX
(0.001,    0.00758,    0.02) ; KM
(0.5,      4.42,          9) ; V1 (L)
(0.00001,  0.000307, 0.0006) ; V2
(0.1,     13,            30) ; Q
(-100,    -3.41223,      50) ; CLALB

$DES DADT(1) = -K10*A(1) -K12*A(1) +K21*A(2) -VMAX*A(1)/(KM*V1 + A(1))
     DADT(2) =            K12*A(1) -K21*A(2)
        cAUC = A(1)/V1
     DADT(3) = cAUC

$ERROR
IPRED = F
LLOQ = 1  ; (mg/L)

; --- Residual error SDs ---
PROP_SD = SIGMA(1)
ADD_SD  = SIGMA(2)

; --- BLQ inflation (gentle, stable) ---
IF (BLQ.EQ.1) ADD_SD = ADD_SD + LLOQ

; --- Final residual error ---
W = SQRT(ADD_SD**2+PROP_SD**2)

IF (W.LE.0.000001) W=0.000001 ; Protective code

IRES=DV-IPRED
IWRES=IRES/W
Y = IPRED * (1 + EPS(1)) + EPS(2)
```

```
$OMEGA 0.366 ; IIV CL

$SIGMA    ; residual variability
0.327     ; EPS(1), proportional
1E-13 FIX ; EPS(2); additive, required by M7+ censoring method

$EST
METHOD=1 INTERACTION; FOCE-I
MAXEVAL=9999
SIG=3
SIGL=3
PRINT=5

$COVARIANCE PRINT=E UNCONDITIONAL MATRIX=S

$TABLE ; output table for standard outcomes
ID TIME TAD DV EVID PRED IPRED WRES IWRES RES IRES CWRES CRES NOPRINT
ONEHEADER FILE=run16_sdtab

$TABLE ; output table for parameters and covariates
ID cAUC CL VMAX KM V1 V2 Q K10 K12 K21 ADA SEX COL BW ALB AGE CREAT NOPRINT
NOAPPEND ONEHEADER FILE=run16_pa_cov
```

A2.2. Two-Compartment Model with Linear Elimination

```
$PROBLEM PMX001 Two Compartment Linear Elimination

$INPUT DUMMY=DROP ID TIME WEEK DOSE=AMT RATE CONC=DROP ADA BW ALB SEX AGE
       CREAT COL EVID BLQ CONC_M7=DV

$DATA dataset_clean.csv IGNORE=@

$ABBR DERIV2=NO

$SUBROUTINES
ADVAN13
TOL=5

$MODEL
NCOMP=3
COMP=(DOSE, DEFDOSE, DEFOBS)
COMP=(PERIPH)
COMP=(AUC)

$PK
IF (EVID.EQ.1) LASTDOSE = TIME
TAD = TIME - LASTDOSE

TVCL = THETA(1)
TVV1 = THETA(2)
TVV2 = THETA(3)
 TVQ = THETA(4)

IF(ADA.EQ.0) CLADA = 1   ; Most common
IF(ADA.EQ.1) CLADA = ( 1 + THETA(5))
V1ALB = ( 1 + THETA(6)*(ALB - 31.84))

  CL = TVCL * EXP(ETA(1)) * CLADA
  V1 = TVV1 * EXP(ETA(2)) * V1ALB
  V2 = TVV2
   Q = TVQ

 K10 = CL/V1
 K12 = Q/V1
 K21 = Q/V2

  S1 = V1/1 ; scale prediction based on DOSE (mg) and DV (mg/L)
  S2 = V2/1
```

```
$THETA ; values are determined in 3 iterations
(0.001,   1.41,      3) ; CL (L/d)
(0.001,   4.73,      7) ; V1 (L)
(0.01,    0.231,   0.8) ; V2
(0.001,   2.15,      5) ; Q
(0.001,   5.78,      8) ; CLADA
(-0.140, -0.127, -0.001) ; V1ALB1

$DES DADT(1) = -K10*A(1) -K12*A(1) +K21*A(2) ; ODE central    compartment
     DADT(2) =            K12*A(1) -K21*A(2) ; ODE peripheral compartment
        cAUC = A(1)/V1                        ; calculates cumulative AUC
     DADT(3) = cAUC                           ; continuous AUC integration

$ERROR
IPRED = F
LLOQ = 1  ; (mg/L)

; --- Residual error SDs ---
PROP_SD = SIGMA(1)
ADD_SD  = SIGMA(2)

; --- BLQ inflation (gentle, stable) ---
IF (BLQ.EQ.1) ADD_SD = ADD_SD + LLOQ

; --- Final residual error ---
W = SQRT(ADD_SD**2+PROP_SD**2)

IF (W.LE.0.000001) W=0.000001 ; Protective code

IRES=DV-IPRED
IWRES=IRES/W
Y = IPRED * (1 + EPS(1)) + EPS(2)

$OMEGA
0.608 ; IIV CL
0.375 ; IIV V1

$SIGMA    ; residual variability
0.289     ; EPS(1), proportional
1E-13 FIX ; EPS(2); additive, required by M7+ censoring method

$EST
METHOD=1 INTERACTION; FOCE-I
MAXEVAL=9999
SIG=3
```

```
PRINT=5

$COVARIANCE PRINT=E UNCONDITIONAL MATRIX=S

$TABLE ; output table for standard outcomes
ID TIME TAD DV cAUC EVID PRED IPRED WRES IWRES RES IRES CWRES CRES NOPRINT
ONEHEADER FILE=run12_sdtab

$TABLE ; output table for parameters and covariates
ID cAUC CL V1 V2 Q K10 K12 K21 ADA SEX COL BW ALB AGE CREAT NOPRINT NOAPPEND
ONEHEADER FILE=run12_pa_cov
```

A3. Exploratory Data Anslysis



Histograms of Continuous Covariates



A4. Exposure-Response Quartile Quartile Plot

```
# A tibble: 2 x 6
  COL              n median_cAUC IQR_cAUC mean_cAUC sd_cAUC
  <fct>        <int>       <dbl>    <dbl>     <dbl>   <dbl>
1 No Colectomy 10539        1.37     32.2      22.9    38.2
2 Colectomy      936       48.4      61.3      56.5    41.7
```



A5. Goodness of Fit Plots

Population Prediction (PRED) vs OBS



Logaritmic scale Population Prediction (PRED) vs OBS



CWRES vs PRED



PRED vs RES



PRED vs CRES



CWRES vs Time after Dose